Pensieve header: A unified verification notebook for the PPSA project.

Continues pensieve://2017-06/ and pensieve://2017-08/.

Prolog

```
Go;
```

```
SetOptions[EvaluationNotebook[],
  NotebookEventActions → {
    {"MenuCommand", "Save"} :→ (
      SetDirectory["C:\\drorbn\\AcademicPensieve\\Projects\\PPSA"];
      Get["MakeVSnips.m"]
     ),
    PassEventsDown → True
   }];
```

Initialization

The “degree carrier / filtration parameter” is ℏ, and all “coupling constants” are proportional to it.

TD

```
$TℏD = 5; $TϵD = 2; ϵ /: ϵ^d_. /; d > $TϵD := 0;
Simp[ℰ_] := Collect[ℰ, _CU | _QU, Expand];
```

DeclareAlgebra

QLImplementation

```
ℏ /: ℏ^d_. /; d > $TℏD := 0;
Unprotect[NonCommutativeMultiply];
NonCommutativeMultiply[x_] := x; 0 ** _ = _ ** 0 = 0;
B[x_, x_] = 0; B[x_, y_] := x ** y - y ** x;
(x_Plus) ** y_ := (# ** y) & /@ x; x_ ** (y_Plus) := (x ** #) & /@ y;
```

QLImplementation

```
DeclareAlgebra[U_Symbol, opts__Rule] := Module[{gp, sr, cp, S, pow,
    gs = Generators /. {opts}, cs = Centrals /. {opts}
  },
  gp = Alternatives @@ gs;
  gp = gp | gp_; (* generators pattern *)
  sr = Thread[gs → Range@Length@gs]; (* sorting rule *)
  cp = Alternatives @@ cs; (* centrals pattern *)
  S[ℰ_] := Collect[ℰ, _U, Expand];
  U_i_[ℰ_] := ℰ /. {t : cp ⧴ t_i, u_U ⧴ Replace[u, x_ ⧴ x_i, 1]};
  B[U@(x_)_i_, U@(y_)_i_] := B[U@x_i, U@y_i] = U_i@B[U@x, U@y];
  B[U@(x_)_i_, U@(y_)_j_] /; i =!= j := 0;
  B[U@y_, U@x_] := S[-B[U@x, U@y]];
  x_ ** U[] := x; U[] ** x_ := x;
  (a_ * x_U) ** (b_ * y_U) := If[a b === 0, 0, S[a b (x ** y)]];
  (a_ * x_U) ** y_ := S[a (x ** y)]; x_ ** (a_ * y_U) := S[a (x ** y)];
  U[xx___, x_] ** U[y_, yy___] := If[OrderedQ[{x, y} /. sr],
    U[xx, x, y, yy], U@xx ** (U@y ** U@x + B[U@x, U@y]) ** U@yy];
  U@{c_. * (l : gp)^n_, r___} /; FreeQ[c, gp] := S[c U@Table[l, {n}] ** U@{r}];
  U@{c_. * l : gp, r___} := S[c U[l] ** U@{r}];
  U@{c_, r___} /; FreeQ[c, gp] := S[c U@{r}];
  U@{} = U[];
  U@{l_Plus, r___} := S[U@{#, r} & /@ l];
  U@{l_, r___} := U@{Expand[l], r};
  𝕆_U[poly_, specs___] := Module[{sp, null, vs, us},
    sp = Replace[{specs}, l_List ⧴ (l → null), {1}];
    vs = Join @@ (First /@ sp);
    us = Join @@ (sp /. (l_ → s_) ⧴ (l /. x_i_ ⧴ x_s));
    S[Total[
      CoefficientRules[
        Normal@Series[poly, {ℏ, 0, $TℏD}], vs] /. (p_ → c_) ⧴ c U@(us^p)
      ]] /. x_null ⧴ x
    ];
  pow[ℰ_, 0] = U[]; pow[ℰ_, n_] := pow[ℰ, n - 1] ** ℰ;
  𝕊_U[ℰ_, ss___Rule] := S@Total[
    CoefficientRules[Normal@Series[ℰ, {ℏ, 0, $TℏD}], First /@ {ss}] /.
      (p_ → c_) ⧴ c NonCommutativeMultiply @@ MapThread[pow, {Last /@ {ss}, p}]]
  ]
```

DeclareMorphism

QLImplementation

```
DeclareMorphism[m_, U_ → V_, ongs_List, oncs_List: {}] := (
  Replace[ongs, (g_ → img_) ⧴ (m[U[g]] = img), {1}];
  m[U[]] = V[];
  m[U[g_i_]] := V_i[m[U@g]];
  m[U[vs__]] := NonCommutativeMultiply @@ (m /@ U /@ {vs});
  m[ℰ_] := Simp[ℰ /. oncs /. u_U ⧴ m[u]]; )
```

Implementing $\mathrm{sl}_2^{\gamma\epsilon}$

CU

```
DeclareAlgebra[CU, Generators → {y, a, x}, Centrals → {t}];
B[CU@a, CU@y] = -γ CU@y; B[CU@x, CU@a] = -γ CU@x;
B[CU@x, CU@y] = 2 ϵ CU@a - t CU[];
```

Verifying associativity on triples of generators:

```
With[{bas = CU /@ {y, a, x}},
 Table[z1 ** (z2 ** z3) - (z1 ** z2) ** z3 // Simp,
  {z1, bas}, {z2, bas}, {z3, bas} ] ]
```

```
{{{0, 0, 0}, {0, 0, 0}, {0, 0, 0}},
 {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}, {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}}
```

Verifying associativity on a “random” triple:

```
With[{z1 = CU[y, y, a, a, x, x], z2 = CU[y, a, x], z3 = CU[y, y, a, x]}, {
   (rhs = (z1 ** z2) ** z3 // Simp) // Short,
   z1 ** (z2 ** z3) - rhs // Simp
  }] // Timing
```

$\{0.46875, \{(28\,t^2\gamma^4 + 116\,t\,\gamma^5\epsilon + 120\,\gamma^6\epsilon^2)\,\mathrm{CU}[y, y, y, x, x] + (4\,t^3\gamma + 8\,t^2\gamma^2\epsilon + 4\,t\,\gamma^3\epsilon^2)\,\mathrm{CU}[y, y, a, a, a, x] + \ll 22 \gg + \mathrm{CU}[y, y, y, y, y, a, a, a, a, x, x, x, x], 0\}\}$

Implementing $\mathcal{U}_{\gamma\epsilon;\hbar}$

With $q = e^{\hbar\gamma\epsilon}$, $A = e^{-\hbar\epsilon a}$, and $[f, g]_q := fg - qgf$, our algebra is $\mathcal{U}_{\gamma\epsilon;\hbar} = \langle t, y, a, x\rangle / \mathcal{R}$, where $\mathcal{R} = ([t, *] = 0,\ [a, y] = -\gamma y,\ [x, y]_q = (1 - T^2 A^2)/\hbar,\ [x, a] = -\gamma x)$.

QU

```
DeclareAlgebra[QU, Generators → {y, a, x}, Centrals → {t, T}];
q := Normal@Series[e^(γ ϵ ℏ), {ℏ, 0, $TℏD}];
B[QU@a, QU@y] = -γ QU@y; B[QU@x, QU@a] = -γ QU@x;
B[QU@x, QU@y] := (q - 1) QU@{y, x} + 𝕆_QU[(1 - T^2 e^(-2 ℏ ϵ a)) / ℏ, {a}];
```

```
With[{bas = QU /@ {y, a, x}}, Table[z1 ** z2 - z2 ** z1 // Simp, {z1, bas}, {z2, bas}] ]
```

$\{\{0, \gamma\,\mathrm{QU}[y], \left(-\frac{1}{\hbar} + \frac{T^2}{\hbar}\right)\mathrm{QU}[\,] - 2\,T^2\epsilon\,\mathrm{QU}[a] + 2\,T^2\epsilon^2\hbar\,\mathrm{QU}[a, a] + \left(-\gamma\epsilon\hbar - \frac{1}{2}\gamma^2\epsilon^2\hbar^2\right)\mathrm{QU}[y, x]\},$
$\{-\gamma\,\mathrm{QU}[y], 0, \gamma\,\mathrm{QU}[x]\},$
$\{\left(\frac{1}{\hbar} - \frac{T^2}{\hbar}\right)\mathrm{QU}[\,] + 2\,T^2\epsilon\,\mathrm{QU}[a] - 2\,T^2\epsilon^2\hbar\,\mathrm{QU}[a, a] + \left(\gamma\epsilon\hbar + \frac{1}{2}\gamma^2\epsilon^2\hbar^2\right)\mathrm{QU}[y, x], -\gamma\,\mathrm{QU}[x], 0\}\}$

Verifying associativity on triples of generators:

```
With[{bas = QU /@ {y, a, x}},
 Table[z1 ** (z2 ** z3) - (z1 ** z2) ** z3 // Simp,
  {z1, bas}, {z2, bas}, {z3, bas} ] ]
```

```
{{{0, 0, 0}, {0, 0, 0}, {0, 0, 0}},
 {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}, {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}}
```

Verifying associativity on a “random” triple (~34 secs):

```
With[{z1 = QU[y, y, a, a, x, x], z2 = QU[y, a, x], z3 = QU[y, y, a, x]}, {
   (rhs = (z1 ** z2) ** z3 // Simp) // Short,
   z1 ** (z2 ** z3) - rhs // Simp
  }] // Timing
```

```
{34., {<<1>>, 0}}
```

Verifying that $\mathrm{Lim}_{\hbar\to 0}$ QU = CU using a “random” product (~23 secs):

```
With[{z1 = CU[y, y, a, a, x, x], z2 = CU[y, a, x], z3 = CU[y, y, a, x]}, {
   Short[lhs = z1 ** (z2 ** z3)],
   Short[rhs = (QU @@ z1) ** ((QU @@ z2) ** (QU @@ z3))],
   Expand[Limit[rhs /. {QU → CU, T → ℯ^(ℏ t/2)}, ℏ → 0] == lhs]
  }] // Timing
```

$\{22.5313, \{2\,(8\,t^2\,\gamma^4 + 16\,t\,\gamma^5\,\epsilon)\,\mathrm{CU}[y,y,y,x,x] + (8\,t\,\gamma^5\,\epsilon + 16\,\gamma^6\,\epsilon^2)\,\mathrm{CU}[y,y,y,x,x] + \ll 93 \gg +$
$\mathrm{CU}[y,y,y,y,y,a,a,a,a,x,x,x,x], (-8\,T^2\,\gamma^6\,\epsilon^2 + 8\,T^4\,\gamma^6\,\epsilon^2)\,\mathrm{QU}[y,y,y,x,x] +$
$\ll 422 \gg + \left(\gamma\,\epsilon\,\hbar + \frac{15}{2}\,\gamma^2\,\epsilon^2\,\hbar^2\right)\mathrm{QU}[y,y,y,y,y,a,a,a,a,x,x,x,x], \mathrm{True}\}\}$

Implementing θ

theta

```
DeclareMorphism[Cθ, CU → CU, {y → -CU@x, a → -CU@a, x → -CU@y}, {t → -t, T → T^-1}];
DeclareMorphism[Qθ, QU → QU,
 {y → 𝕆_QU[-T^-1 ℯ^(ℏ ϵ a) x, {a, x}], a → -QU@a, x → 𝕆_QU[-T^-1 ℯ^(ℏ ϵ a) y, {a, y}]}, {t → -t, T → T^-1}]
```

Verifying involutivity on CU:

```
With[{bas = CU /@ {y, a, x}},
 Table[z → Cθ[z] → Cθ[Cθ[z]], {z, bas}] ]
```

```
{CU[y] → -CU[x] → CU[y], CU[a] → -CU[a] → CU[a], CU[x] → -CU[y] → CU[x]}
```

Verifying that θ is a multiplicative homomorphism on CU:

```
With[{bas = CU /@ {y, a, x}},
 Table[Cθ[z1 ** z2] - Cθ[z1] ** Cθ[z2], {z1, bas}, {z2, bas}] ]
```

```
{{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}
```

Verifying involutivity on QU:

```
With[{bas = QU /@ {y, a, x}},
 Table[z → Qθ[z] → Qθ[Qθ[z]], {z, bas}] ]
```

$$\left\{\mathrm{QU}[y] \to -\frac{\mathrm{QU}[x]}{T} - \frac{\epsilon\hbar\,\mathrm{QU}[a,x]}{T} - \frac{\epsilon^2\hbar^2\,\mathrm{QU}[a,a,x]}{2T} \to \mathrm{QU}[y],\ \mathrm{QU}[a] \to -\mathrm{QU}[a] \to \mathrm{QU}[a],\right.$$
$$\left.\mathrm{QU}[x] \to \left(-\frac{1}{T} + \frac{\gamma\epsilon\hbar}{T} - \frac{\gamma^2\epsilon^2\hbar^2}{2T}\right)\mathrm{QU}[y] + \left(-\frac{\epsilon\hbar}{T} + \frac{\gamma\epsilon^2\hbar^2}{T}\right)\mathrm{QU}[y,a] - \frac{\epsilon^2\hbar^2\,\mathrm{QU}[y,a,a]}{2T} \to \mathrm{QU}[x]\right\}$$

Verifying that θ is a multiplicative homomorphism on QU:

```
With[{bas = QU /@ {y, a, x}},
 Table[{z1, z2} → Simp[Qθ[z1 ** z2] - Qθ[z1] ** Qθ[z2]], {z1, bas}, {z2, bas}] ]
```

```
{{{QU[y], QU[y]} → 0, {QU[y], QU[a]} → 0, {QU[y], QU[x]} → 0},
 {{QU[a], QU[y]} → 0, {QU[a], QU[a]} → 0, {QU[a], QU[x]} → 0},
 {{QU[x], QU[y]} → 0, {QU[x], QU[a]} → 0, {QU[x], QU[x]} → 0}}
```

The Asymmetric Dequantizator

```
DeclareMorphism[ADeq, QU → CU, {y → CU@y, a → CU@a, x → CU@x}]
```

```
ADeq[QU[y, y, a, a, x, x]]
```

```
CU[y, y, a, a, x, x]
```

$$g = \frac{\mathrm{Cosh}\left[\hbar\left(\frac{\epsilon - t}{2} + \epsilon\, a\right)\right] - \mathrm{Cosh}\left[\hbar\sqrt{\left(\frac{t-\epsilon}{2}\right)^2 + \epsilon\,\omega}\right]}{\hbar\,\mathrm{Sinh}\left[\frac{-\epsilon\hbar}{2}\right]\left(\omega - \epsilon\, a^2 + (t-\epsilon)\, a\right)};$$

```
Series[g, {ℏ, 0, 2}]
```

$$1 + \left(\left(-a\,t^3 + 3\,a\,t^2\,\epsilon + 3\,a^2\,t^2\,\epsilon - t^2\,\omega + 2\,t\,\epsilon\,\omega - 2\,\epsilon\,\omega^2\right)\hbar^2\right) \Big/ \left(24\left(-a\,t + a\,\epsilon + a^2\,\epsilon - \omega\right)\right) + O[\hbar]^3$$

$$r = 2\left(\left(\mathrm{Cosh}\left[h\left(\frac{\epsilon - t}{2} + \epsilon\, a\right)\right] - \mathrm{Cosh}\left[h\,\mathrm{Sqrt}\left[\left(\frac{1}{4}(-t+\epsilon)^2 + \epsilon\,\omega\right)\right]\right]\right)\Big/\right.$$
$$\left.\left(h\,2\,\mathrm{Sinh}\left[-\frac{\epsilon}{2}h\right]\left(\omega - \left(\epsilon\, a^2 - (t-\epsilon)\, a\right)\right)\right)\right);$$

```
Series[r, {h, 0, 2}] // Simplify
```

$$1 - \left(\left(a\,t^2\,(t - 3\,\epsilon) - 3\,a^2\,t^2\,\epsilon + \omega\left(t^2 - 2\,t\,\epsilon + 2\,\epsilon\,\omega\right)\right)h^2\right) \Big/ \left(24\left(a^2\,\epsilon + a\,(-t+\epsilon) - \omega\right)\right) + O[h]^3$$

```
Simplify[g == r /. h → ℏ]
```

```
True
```

$$f = e^{\hbar\,(t/2-\epsilon-\epsilon\, a)}\,\frac{\mathrm{Cosh}\left[\hbar\left(\frac{\epsilon-t}{2}+\epsilon\, a\right)\right]-\mathrm{Cosh}\left[\hbar\sqrt{\left(\frac{t-\epsilon}{2}\right)^2+\epsilon\,\omega}\right]}{\hbar\,\mathrm{Sinh}\left[\frac{-\epsilon\,\hbar}{2}\right]\left(\omega-\epsilon\, a^2+(t-\epsilon)\, a\right)}\, y$$

$$-\left(\left(e^{\left(\frac{t}{2}-\epsilon-a\,\epsilon\right)\hbar}\, y\left(\mathrm{Cosh}\left[\left(a\,\epsilon+\frac{1}{2}(-t+\epsilon)\right)\hbar\right]-\mathrm{Cosh}\left[\sqrt{\frac{1}{4}(t-\epsilon)^2+\epsilon\,\omega}\;\hbar\right]\right)\mathrm{Csch}\left[\frac{\epsilon\,\hbar}{2}\right]\right)\Big/\left(\left(a\,(t-\epsilon)-a^2\,\epsilon+\omega\right)\hbar\right)\right)$$

FullSimplify[f]

$$-\left(\left(e^{\frac{1}{2}(t-2\,(1+a)\,\epsilon)\,\hbar}\, y\left(\mathrm{Cosh}\left[\frac{1}{2}(-t+\epsilon+2\,a\,\epsilon)\,\hbar\right]-\mathrm{Cosh}\left[\sqrt{\frac{1}{4}(t-\epsilon)^2+\epsilon\,\omega}\;\hbar\right]\right)\mathrm{Csch}\left[\frac{\epsilon\,\hbar}{2}\right]\right)\Big/\left(\left(a\,(t-(1+a)\,\epsilon)+\omega\right)\hbar\right)\right)$$

FullSimplify[f // TrigToExp]

$$\left(2\, e^{\frac{1}{2}(t-(1+2\,a)\,\epsilon)\,\hbar}\, y\left(-\mathrm{Cosh}\left[\frac{1}{2}\left(t-(1+2\,a)\,\epsilon\right)\hbar\right]+\mathrm{Cosh}\left[\frac{1}{2}\sqrt{(t-\epsilon)^2+4\,\epsilon\,\omega}\;\hbar\right]\right)\right)\Big/\left(\left(-1+e^{\epsilon\,\hbar}\right)\left(a\,(t-(1+a)\,\epsilon)+\omega\right)\hbar\right)$$

Series[f / y, {ℏ, 0, 3}] // Normal // Simplify

$$\frac{1}{24}\Bigg(24+12\,(t-2\,(1+a)\,\epsilon)\,\hbar-\frac{1}{a^2\,\epsilon+a\,(-t+\epsilon)-\omega}\,2\,\Big(12\,a^3\,t\,\epsilon^2+3\,a^2\,\epsilon\left(-3\,t^2+8\,t\,\epsilon+2\,\epsilon\,\omega\right)+a\left(2\,t^3-9\,t^2\,\epsilon+6\,t\,\epsilon\,(2\,\epsilon-\omega)+12\,\epsilon^2\,\omega\right)+\omega\left(2\,t^2-7\,t\,\epsilon+\epsilon\,(6\,\epsilon+\omega)\right)\Big)\hbar^2-\frac{1}{a^2\,\epsilon+a\,(-t+\epsilon)-\omega}\Big(12\,a^3\,t^2\,\epsilon^2+6\,a^2\,t\,\epsilon\left(-t^2+4\,t\,\epsilon+\epsilon\,\omega\right)+a\left(t^4-6\,t^3\,\epsilon+4\,t^2\,\epsilon\,(3\,\epsilon-\omega)+14\,t\,\epsilon^2\,\omega-2\,\epsilon^2\,\omega^2\right)+\omega\left(t^3-5\,t^2\,\epsilon-2\,\epsilon^2\,\omega+t\,\epsilon\,(8\,\epsilon+\omega)\right)\Big)\hbar^3\Bigg)$$

Collect[Series[f / y, {ℏ, 0, 3}, {ϵ, 0, 2}] // Normal, {ℏ, ϵ}, Simplify]

$$1+\left(\frac{t}{2}+(-1-a)\,\epsilon\right)\hbar+\left(\frac{t^2}{6}+\frac{1}{12}\,\epsilon\,(-7\,(1+a)\,t+\omega)+\frac{(1+a)\,\epsilon^2\,(5\,a\,t+5\,a^2\,t+6\,\omega+7\,a\,\omega)}{12\,(a\,t+\omega)}\right)\hbar^2+\left(\frac{t^3}{24}+\frac{1}{24}\,t\,\epsilon\,(-5\,(1+a)\,t+\omega)+\left((1+a)\,\epsilon^2\left(7\,a^2\,t^2+2\,(4\,t-\omega)\,\omega+7\,a\,t\,(t+\omega)\right)\right)\Big/\left(24\,(a\,t+\omega)\right)\right)\hbar^3$$

$$\text{FullSimplify}\left[\frac{(1+a)\,\epsilon^2\,(5\,a\,t+5\,a^2\,t+6\,\omega+7\,a\,\omega)}{12\,(a\,t+\omega)}\right]$$

$$\frac{(1+a)\,\epsilon^2\,(5\,a\,(1+a)\,t+(6+7\,a)\,\omega)}{12\,(a\,t+\omega)}$$

$$\mathbb{S}_{CU}\left[e^{\hbar\,x},\,x\to CU[a,x]\right]$$

$$CU[\,]+\hbar\,CU[a,x]-\frac{1}{2}\gamma\,\hbar^2\,CU[a,x,x]+\frac{1}{2}\hbar^2\,CU[a,a,x,x]+\frac{1}{3}\gamma^2\,\hbar^3\,CU[a,x,x,x]-\frac{1}{2}\gamma\,\hbar^3\,CU[a,a,x,x,x]-\frac{1}{4}\gamma^3\,\hbar^4\,CU[a,x,x,x,x]+\frac{1}{6}\hbar^3\,CU[a,a,a,x,x,x]+\frac{11}{24}\gamma^2\,\hbar^4\,CU[a,a,x,x,x,x]+\frac{1}{5}\gamma^4\,\hbar^5\,CU[a,x,x,x,x,x]-\frac{1}{4}\gamma\,\hbar^4\,CU[a,a,a,x,x,x,x]-\frac{5}{12}\gamma^3\,\hbar^5\,CU[a,a,x,x,x,x,x]+\frac{1}{24}\hbar^4\,CU[a,a,a,a,x,x,x,x]+\frac{7}{24}\gamma^2\,\hbar^5\,CU[a,a,a,x,x,x,x,x]-\frac{1}{12}\gamma\,\hbar^5\,CU[a,a,a,a,x,x,x,x,x]+\frac{1}{120}\hbar^5\,CU[a,a,a,a,a,x,x,x,x,x]$$